# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Thomas Dziedzic <gostrc@gmail.com>

_hkgname=monoid-subclasses
pkgname=haskell-monoid-subclasses
pkgver=1.2.2
pkgrel=8
pkgdesc="Subclasses of Monoid"
url="https://github.com/blamario/monoid-subclasses/"
license=('BSD')
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-commutative-semigroups' 'haskell-primes' 'haskell-vector')
makedepends=('ghc' 'haskell-quickcheck' 'haskell-quickcheck-instances' 'haskell-tasty'
             'haskell-tasty-quickcheck')
source=("https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz")
sha512sums=('541595c4b4bc0f7e1f4f9fcd032a99770ff9a55e4320b998819cd1193ea5cd94ee7ade6c24b597f0e41f94e1c263270bf9498e581da5a931aac1a99209ecd156')

build() {
    cd $_hkgname-$pkgver
    
    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
    cd $_hkgname-$pkgver
    runhaskell Setup test --show-details=direct
}

package() {
    cd $_hkgname-$pkgver
    
    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 "BSD3-LICENSE.txt" "${pkgdir}/usr/share/licenses/${pkgname}/BSD3-LICENSE.txt"
    rm -f "${pkgdir}/usr/share/doc/${pkgname}/BSD3-LICENSE.txt"
}
